
01 Définition

EN UNE PHRASE

Une référence est l'adresse d'une cellule ou d'une plage dans une formule, par exemple B2  ou

A2:A50 . Selon qu'elle porte ou non le signe $ , elle se décale ou reste fixe quand on recopie la

formule.

Une plage s'écrit avec deux-points entre la première et la dernière cellule : A2:D10  désigne le rectangle de A2 à

D10. Une colonne entière s'écrit B:B , une ligne entière 3:3 .

02 Relative, absolue, mixte

Écriture Type Quand on recopie la formule

B2 Relative La ligne et la colonne se décalent

$B$2 Absolue Rien ne bouge

$B2 Mixte La colonne reste fixe, la ligne se décale

B$2 Mixte La ligne reste fixe, la colonne se décale

💡 ASTUCE

La touche F4

Dans la barre de formule, cliquez sur une référence et appuyez sur F4  : Excel ajoute les $  à votre place. Chaque

appui passe à l'étape suivante : $B$2 , puis B$2 , puis $B2 , puis B2 .

FICHE DE COURS GRATUITE

FORMULES EXCEL | BASES

Références et noms

Une formule ne contient presque jamais de valeurs en dur :

elle désigne des cellules. Savoir comment Excel interprète

ces adresses quand on recopie une formule, et savoir les

remplacer par des noms parlants, c'est la base de toutes les

autres fiches.

Version V1.0

Dernière mise à

jour

03 La référence relative, le comportement par défaut

Une référence relative est mémorisée comme un déplacement : « la cellule deux colonnes à gauche, sur la même

ligne ». C'est ce qui permet d'écrire une formule une seule fois puis de la recopier vers le bas.

Montant = prix x quantité, recopié vers le bas

A B C D

1 Produit Prix Quantité Montant

2 Clavier 49 € 3 =B2*C2

3 Souris 25 € 10 =B3*C3

4 Écran 189 € 2 =B4*C4

04 La référence absolue : bloquer une cellule

Quand toutes les lignes doivent utiliser la même cellule, par exemple un taux de TVA, il faut la bloquer avec $ .
Sans les dollars, la formule recopiée irait chercher F2, puis F3, puis F4, qui sont vides.

Prix TTC avec un taux unique en F1

A B C D E F

1 Produit Prix HT Prix TTC Taux TVA 20 %

2 Clavier 49 € =B2*(1+$F$1)

3 Souris 25 € =B3*(1+$F$1)

4 Écran 189 € =B4*(1+$F$1)

05 La référence mixte : une table de calcul

La référence mixte bloque seulement la ligne ou seulement la colonne. Elle sert quand une formule est recopiée à

la fois vers le bas et vers la droite, comme une table de multiplication : les valeurs de ligne sont en colonne A,

celles de colonne en ligne 1.

UNE SEULE FORMULE EN B2, RECOPIÉE SUR TOUT LE TABLEAU

=$A2*B$1

$A2  reste toujours en colonne A, B$1  reste toujours en ligne 1. Une seule formule remplit tout le tableau.

06 Faire référence à une autre feuille ou un autre classeur

Écriture Ce qu'elle désigne

=Janvier!B2 La cellule B2 de la feuille Janvier

='Ventes 2026'!B2 Le nom de feuille contient un espace : apostrophes obligatoires

=SOMME(Janvier:Mars!B2) B2 additionnée sur toutes les feuilles de Janvier à Mars

='[Budget.xlsx]Janvier'!B2 La cellule B2 d'un autre classeur ouvert

Le plus simple reste de taper = , puis de cliquer sur l'autre feuille et sur la cellule : Excel écrit la référence correc‐

tement, apostrophes comprises.

07 Nommer une cellule ou une plage

Un nom remplace une adresse par un mot. =B2*(1+Taux_TVA)  se lit tout de suite, alors que =B2*(1+$F$1)
oblige à aller voir ce que contient F1. Un nom est toujours absolu : pas besoin de $ .

Trois autres façons de faire :

Formules > Définir un nom : permet de choisir la portée et d'ajouter un commentaire.

Ctrl + Maj + F3  (Créer à partir de la sélection) : nomme chaque colonne d'après son en-tête, en une fois.

Ctrl + F3  (Gestionnaire de noms) : liste tous les noms du classeur, pour les modifier ou les supprimer.

Dans une formule, F3  affiche la liste des noms pour en coller un sans faute de frappe.

UN NOM PEUT AUSSI CONTENIR UNE CONSTANTE OU UNE FORMULE

Formules > Définir un nom, Nom : Taux_TVA, Fait référence à : =20%
=B2*(1+Taux_TVA)

💡 ASTUCE

Choisir la portée

Un nom de portée Classeur est utilisable sur toutes les feuilles. Un nom de portée Feuille n'existe que sur sa feuille

: pratique quand chaque feuille mensuelle a son propre Objectif .

08 Les tableaux structurés

Un tableau Excel (Accueil > Mettre sous forme de tableau, ou Ctrl + L ) se nomme automatiquement, par

exemple Ventes , et chaque colonne devient adressable par son en-tête. Le tableau s'agrandit tout seul quand
on ajoute une ligne, et les formules suivent.

Écriture Ce qu'elle désigne

Ventes[Montant] Toute la colonne Montant, sans l'en-tête

[@Montant] La valeur de Montant sur la ligne de la formule

Ventes[#En-têtes] La ligne d'en-têtes

Ventes[#Tout] Tout le tableau, en-têtes compris

Ventes[[Prix]:[Quantité]] Les colonnes de Prix à Quantité

Sélectionnez la cellule ou la plage.1

Cliquez dans la zone Nom, à gauche de la barre de formule.2

Tapez le nom, par exemple Taux_TVA , puis validez avec Entrée .3

Tableau « Ventes » : la formule de la colonne Montant se remplit seule

A B C D

1 Produit Prix Quantité Montant

2 Clavier 49 € 3 =[@Prix]*[@Quantité]

3 Souris 25 € 10 250 €

4 Écran 189 € 2 378 €

UTILISER LA COLONNE D'UN TABLEAU AILLEURS DANS LE CLASSEUR

=SOMME(Ventes[Montant])
=NB.SI(Ventes[Produit];"Clavier")

Renommez le tableau dès sa création : Création de tableau > Nom du tableau. Ventes  est plus clair que

Tableau1 .

09 Pièges à éviter

⚠️ PIÈGE

Oublier le $ avant de recopier

La première ligne donne le bon résultat, les suivantes renvoient 0 ou une erreur : la cellule de référence a « glissé ».

Cliquez sur une cellule recopiée et regardez la formule : la référence fautive saute aux yeux.

⚠️ PIÈGE

Un nom qui ressemble à une adresse

TVA1  ou CA2026  sont refusés comme noms : ce sont des adresses de cellules valides (colonne TVA, ligne 1). Pré‐

férez Taux_TVA  ou CA_2026 . Pas d'espace non plus : utilisez le tiret bas.

⚠️ PIÈGE

Des noms orphelins

Supprimer une ligne ou une feuille ne supprime pas le nom qui pointait dessus : il affiche alors #REF! . Faites le

ménage de temps en temps dans le gestionnaire de noms ( Ctrl + F3 ).

⚠️ PIÈGE

Lier des classeurs sans le savoir

Copier une feuille ou une formule depuis un autre classeur peut créer des liaisons externes, signalées à chaque ou‐

verture. Pour les retrouver : onglet Données, bouton Modifier les liens (son nom varie légèrement selon la version

d'Excel).

01 Définition

EN UNE PHRASE

L'assistant fonction est la fenêtre qui aide à choisir une fonction et à remplir ses arguments, un

champ par argument, avec une description et le résultat calculé en direct.

Il s'ouvre en cliquant sur le bouton fx , à gauche de la barre de formule, ou avec le raccourci Maj + F3 .

02 Trouver la bonne fonction

💡 ASTUCE

Par catégorie

Le menu Sélectionnez une catégorie range les fonctions par familles : Logique, Texte, Date et heure, Recherche et

référence... C'est un bon moyen de découvrir ce qu'Excel sait faire. On retrouve les mêmes familles dans l'onglet

Formules, groupe Bibliothèque de fonctions.

03 La fenêtre des arguments

Chaque argument a son champ. Les arguments en gras sont obligatoires, les autres sont facultatifs. En cliquant

dans un champ, sa description s'affiche en bas.

À droite de chaque champ, Excel affiche la valeur de l'argument : la vérification est immédiate.

FICHE DE COURS GRATUITE

FORMULES EXCEL | BASES

L'assistant fonction

Personne ne connaît par cœur les arguments des 500

fonctions d'Excel. L'assistant fonction les présente un par

un, avec une explication et un aperçu du résultat. C'est

l'outil idéal pour découvrir une fonction, puis pour

comprendre une formule écrite par quelqu'un d'autre.

Sélectionnez la cellule qui recevra le résultat.1

Cliquez sur fx  ou appuyez sur Maj + F3  : la fenêtre Insérer une fonction s'ouvre.2

Décrivez ce que vous voulez faire dans Recherchez une fonction, par exemple « compter » ou « date », puis

cliquez sur OK : Excel propose une liste.

3

Cliquez sur une fonction : sa syntaxe et une phrase d'explication s'affichent en bas de la fenêtre.4

Validez : la fenêtre des arguments s'ouvre.5

Version V1.0

Dernière mise à

jour
30/10/2026

Niveau 1 / 3, Découverte

Durée de lecture 7 min

Auteur Clément Delobel

En bas, Résultat = montre le résultat de la formule avant même de valider.

Pour désigner une plage, cliquez dans le champ puis sélectionnez les cellules directement sur la feuille. La
fenêtre se replie si besoin.

Les guillemets autour d'un texte et les points-virgules entre arguments sont ajoutés par Excel.

💡 ASTUCE

Rouvrir l'assistant sur une formule existante

Cliquez dans une formule déjà écrite, juste après le nom d'une fonction, puis appuyez sur Maj + F3  ou cliquez

sur fx  : la fenêtre des arguments s'ouvre, remplie. C'est la façon la plus simple de relire une formule.

04 Sans fenêtre : la saisie semi-automatique

En tapant =REC , Excel propose une liste de fonctions. Flèches pour choisir, Tab  pour valider. Une fois la paren‐

thèse ouverte, une info-bulle affiche la syntaxe et met en gras l'argument en cours. Cliquer sur le nom d'un argu‐

ment dans l'info-bulle sélectionne la partie correspondante de la formule.

L'INFO-BULLE AFFICHÉE EN TAPANT LA FORMULE

=RECHERCHEV(valeur_cherchée; table_matrice; no_index_col; [valeur_proche])

Les crochets signalent un argument facultatif.

05 Comprendre une formule existante

Trois outils pour décortiquer une formule qui ne donne pas le résultat attendu :

⚠️ PIÈGE

F9 puis Entrée

Après F9 , si vous validez avec Entrée  au lieu d' Échap , la partie de formule est remplacée définitivement par

sa valeur. Si cela arrive, Ctrl + Z  annule tout de suite.

06 Pièges à éviter

⚠️ PIÈGE

Des points-virgules, pas des virgules

Dans Excel en français, les arguments sont séparés par des points-virgules : =SOMME(A1;B1) . Une formule copiée

depuis un site anglais, avec des virgules, renvoie une erreur ou un résultat faux.

F9 sur une partie de formule : dans la barre de formule, sélectionnez un morceau, par exemple B2*C2 , puis
appuyez sur F9 . Excel le remplace par sa valeur. Appuyez ensuite sur Échap  pour ne rien modifier.

1

L'évaluation de formule : Formules > Évaluation de formule. Cliquez sur Évaluer pour calculer la formule

étape par étape, en commençant par la partie la plus imbriquée.

2

Repérer les antécédents : Formules > Repérer les antécédents dessine des flèches vers les cellules

utilisées par la formule.

3

⚠️ PIÈGE

Le nom traduit des fonctions

VLOOKUP s'appelle RECHERCHEV, IF s'appelle SI, SUMIFS s'appelle SOMME.SI.ENS. Les noms anglais ne sont pas

reconnus dans un Excel en français. En cas de doute, la recherche de l'assistant fonction accepte une description

en français.

⚠️ PIÈGE

Tout taper dans un seul champ

Dans la fenêtre des arguments, une formule entière collée dans le premier champ devient un seul argument. Rem‐

plissez un champ par argument, ou fermez la fenêtre et tapez la formule directement dans la cellule.

01 Définition

EN UNE PHRASE

SOMME additionne, MOYENNE calcule la moyenne, MIN et MAX renvoient la plus petite et la plus

grande valeur. Toutes les quatre ne prennent en compte que les nombres : le texte et les cellules

vides sont ignorés.

02 Syntaxe

SYNTAXE

=SOMME(nombre1; [nombre2]; ...)
=MOYENNE(nombre1; [nombre2]; ...)
=MIN(nombre1; [nombre2]; ...)
=MAX(nombre1; [nombre2]; ...)

Chaque argument peut être une cellule, une plage ou un nombre. On peut en donner jusqu'à 255.

FICHE DE COURS GRATUITE

FORMULES EXCEL | BASES

SOMME, MOYENNE,
MIN, MAX

Ce sont les premières fonctions que l'on apprend, et

pourtant elles cachent quelques surprises : cellules vides,

zéros, nombres saisis comme du texte. Cette fiche fait le

tour des quatre calculs de base et de ce qui peut les

fausser.

03 Exemples

Chiffre d'affaires par commercial

A B

1 Commercial CA

2 Alice 12 400 €

3 Bruno 8 900 €

4 Chloé 15 200 €

5 Total =SOMME(B2:B4)

6 Moyenne =MOYENNE(B2:B4)

7 Meilleur =MAX(B2:B4)

PLUSIEURS PLAGES, OU DES CELLULES ISOLÉES

=SOMME(B2:B4;D2:D4)
=SOMME(B2;B4;1000)
=MAX(Janvier!B2:B4;Février!B2:B4)

04 La somme automatique

Placez-vous sous une colonne de nombres et appuyez sur Alt + =  (ou Accueil > Somme automatique). Excel
propose une formule SOMME  sur la plage au-dessus : vérifiez-la, puis validez avec Entrée . La petite flèche du

bouton donne aussi accès à Moyenne, Max, Min et Nombre.

💡 ASTUCE

Sans formule : la barre d'état

Sélectionnez simplement des cellules : la somme, la moyenne et le nombre de valeurs s'affichent en bas de la fe‐

nêtre. Un clic droit sur la barre d'état ajoute le minimum et le maximum.

05 Pièges à éviter

⚠️ PIÈGE

Cellule vide ou zéro : pas la même moyenne

MOYENNE  ignore les cellules vides mais compte les zéros. Avec les notes 12, 14 et une cellule vide, la moyenne vaut

13. Avec 12, 14 et 0, elle vaut 8,67. Laissez vide ce qui n'est pas encore connu.

⚠️ PIÈGE

Des nombres stockés en texte

Un nombre importé d'un logiciel est parfois du texte : il est aligné à gauche et marqué d'un petit triangle vert.

SOMME  l'ignore sans prévenir, le total est donc faux. Sélectionnez les cellules, cliquez sur le losange d'avertisse‐

ment puis sur Convertir en nombre.

⚠️ PIÈGE

Une plage qui ne suit pas

Si vous insérez une ligne juste sous la plage additionnée, elle n'est pas forcément ajoutée au total. Avec un tableau

structuré, =SOMME(Ventes[CA])  s'agrandit automatiquement.

⚠️ PIÈGE

Additionner une liste filtrée

SOMME  additionne aussi les lignes masquées par un filtre. Pour ne totaliser que les lignes visibles, utilisez

=SOUS.TOTAL(109;B2:B50) . C'est d'ailleurs la formule qu'Excel insère dans la ligne Total d'un tableau.

01 Définition

EN UNE PHRASE

Un pourcentage est une fraction de 100 : 25 % vaut 0,25. Dans Excel, on calcule la fraction avec une

formule, puis on l'affiche en pourcentage avec le format % ( Ctrl + Maj + %  ou Accueil > Style de

pourcentage).

Le format ne change pas la valeur : la cellule contient 0,25, elle affiche 25 %.

02 Les cinq formules à connaître

Calcul Formule Exemple

Part d'un total =partie/total =B2/B$6

Taux d'évolution =(arrivée-départ)/départ =(C2-B2)/B2

Appliquer une hausse =valeur*(1+taux) =B2*(1+5%)

Appliquer une remise =valeur*(1-taux) =B2*(1-15%)

Retrouver le HT d'un TTC =TTC/(1+taux) =B2/(1+20%)

03 La part d'un total

Chaque ligne est divisée par le total. Le total est bloqué avec un $  devant la ligne pour que la formule se recopie

sans glisser (voir les références).

FICHE DE COURS GRATUITE

FORMULES EXCEL | BASES

Calculer un
pourcentage
Excel n'a pas de fonction POURCENTAGE : tout se fait avec

une division ou une multiplication, puis le format

pourcentage. Encore faut-il savoir quelle formule écrire.

Voici les cinq calculs que l'on rencontre partout, du tableau

de ventes à la facture.

Part de chaque région dans le chiffre d'affaires

A B C

1 Région CA Part

2 Nord 42 000 € =B2/B$6

3 Sud 31 500 € 25 %

4 Est 18 900 € 15 %

5 Ouest 33 600 € 27 %

6 Total 126 000 € 100 %

04 Le taux d'évolution

On compare toujours à la valeur de départ. Un résultat négatif est une baisse.

Évolution du CA entre 2025 et 2026

A B C D

1 Produit 2025 2026 Évolution

2 Formation 80 000 € 92 000 € =(C2-B2)/B2

3 Conseil 45 000 € 40 500 € -10 %

ÉCRITURE ÉQUIVALENTE, PLUS COURTE

=C2/B2-1

05 Hausse, remise, HT et TTC

APPLIQUER UN TAUX

Prix augmenté de 5 %
=B2*(1+5%)
Prix après une remise de 15 %
=B2*(1-15%)
Prix TTC à partir du HT (TVA à 20 %)
=B2*(1+20%)
Prix HT à partir du TTC
=B2/(1+20%)
Montant de la TVA à partir du TTC
=B2-B2/(1+20%)

💡 ASTUCE

Mettre le taux dans une cellule

Plutôt que d'écrire 20%  dans chaque formule, saisissez le taux dans une cellule nommée, par exemple Taux_TVA ,

puis écrivez =B2*(1+Taux_TVA) . Le jour où le taux change, une seule cellule est à modifier.

06 Pièges à éviter

⚠️ PIÈGE

Enlever 20 % n'est pas retrouver le HT

Un prix TTC de 120 € correspond à 100 € HT. Or =120*(1-20%)  donne 96 €. Pour retrouver le HT, il faut diviser :

=120/(1+20%) .

⚠️ PIÈGE

Le format % multiplie l'affichage par 100

Une cellule qui contient 25 et à laquelle on applique le format % affiche 2 500 %. Soit on divise par 100, soit on ap‐

plique le format % avant de saisir : Excel interprète alors 25 comme 25 %.

⚠️ PIÈGE

Points de pourcentage ou pourcentage

Un taux qui passe de 10 % à 12 % augmente de 2 points, mais de 20 % en valeur relative. Précisez toujours lequel

des deux vous affichez.

⚠️ PIÈGE

Division par zéro

Un taux d'évolution sur une valeur de départ vide ou nulle renvoie #DIV/0! . Protégez la formule :

=SI(B2=0;"";C2/B2-1) .

01 Définition

EN UNE PHRASE

La fonction SI teste une condition et renvoie une valeur si la condition est vraie, une autre si elle est

fausse.

02 Syntaxe

SYNTAXE

=SI(test_logique; valeur_si_vrai; [valeur_si_faux])

Argument Rôle

test_logique Une comparaison qui vaut VRAI ou FAUX, par exemple B2>=10

valeur_si_vrai Ce qu'affiche la cellule si le test est vrai

valeur_si_faux Ce qu'affiche la cellule sinon. Facultatif, mais à toujours renseigner

03 Les opérateurs de comparaison

Opérateur Signification Exemple

= égal à B2="Paris"

<> différent de B2<>"Annulé"

> et < supérieur, inférieur B2>1000

>= et <= supérieur ou égal, inférieur ou égal B2>=10

FICHE DE COURS GRATUITE

FORMULES EXCEL | LOGIQUE

La fonction SI
SI est la fonction qui fait réfléchir une feuille de calcul : si le

CA dépasse l'objectif, afficher « Atteint », sinon « En retard

». Elle est simple, elle est partout, et elle sert de brique à

des formules bien plus riches.

04 Exemples

Résultat d'examen

A B C

1 Candidat Note Résultat

2 Inès 14 =SI(B2>=10;"Admis";"Ajourné")

3 Marc 8,5 Ajourné

4 Zoé 10 Admis

CALCULER UNE PRIME DE 5 % AU-DELÀ DE 1 000 € DE VENTES

=SI(B2>1000;B2*5%;0)

COMPARER À UNE AUTRE CELLULE

=SI(C2>=D2;"Objectif atteint";"En retard")

LAISSER LA CELLULE VIDE TANT QUE RIEN N'EST SAISI

=SI(B2="";"";B2*1,2)

Les deux guillemets collés ""  représentent un texte vide : la cellule paraît vide.

05 Pièges à éviter

⚠️ PIÈGE

Des guillemets autour d'un nombre

=SI(B2>="10";...)  compare B2 au texte « 10 », et le résultat est faux. Les nombres s'écrivent sans guillemets,

les textes toujours entre guillemets.

⚠️ PIÈGE

Oublier le troisième argument

=SI(B2>=10;"Admis")  affiche FAUX quand la condition n'est pas remplie. Indiquez toujours la valeur si faux, quitte

à mettre "" .

⚠️ PIÈGE

Les majuscules

SI ne fait pas la différence entre majuscules et minuscules : B2="paris"  est vrai si B2 contient « PARIS ». Pour

une comparaison stricte, utilisez =SI(EXACT(B2;"Paris");...) .

⚠️ PIÈGE

Trop de SI les uns dans les autres

Au-delà de deux ou trois conditions, la formule devient illisible. Passez à SI.CONDITIONS ou à une table de corres‐

pondance avec RECHERCHEX.

01 Définition

EN UNE PHRASE

Un SI imbriqué place un nouveau SI dans l'argument « valeur si faux » du précédent. SI.CONDITIONS

fait la même chose en une seule fonction, en listant les paires condition ; résultat dans l'ordre.

02 Les SI imbriqués

Mention selon la note

A B C

1 Élève Note Mention

2 Léa 17 =SI(B2>=16;"Très bien";SI(B2>=14;"Bien";SI(B2>=12;"Assez bien";"Sans mention")))

3 Hugo 13 Assez bien

4 Nora 9 Sans mention

La lecture se fait de gauche à droite : si la note est au moins 16, « Très bien » ; sinon, si elle est au moins 14, «

Bien » ; et ainsi de suite. Le dernier argument est le cas par défaut.

💡 ASTUCE

L'ordre des tests compte

Excel s'arrête au premier test vrai. Testez donc du plus exigeant au moins exigeant. Si l'on testait B2>=12  en pre‐

mier, une note de 17 recevrait « Assez bien ».

FICHE DE COURS GRATUITE

FORMULES EXCEL | LOGIQUE

SI imbriqués et
SI.CONDITIONS
Une note devient une mention, un CA devient un taux de

commission, un âge devient une tranche. Dès qu'il y a plus

de deux résultats possibles, il faut enchaîner les tests. Deux

techniques existent : imbriquer des SI, ou utiliser

SI.CONDITIONS, plus lisible.

03 SI.CONDITIONS

SYNTAXE

=SI.CONDITIONS(test1; valeur1; [test2; valeur2]; ...)

LA MÊME MENTION, PLUS LISIBLE

=SI.CONDITIONS(B2>=16;"Très bien";B2>=14;"Bien";B2>=12;"Assez bien";VRAI;"Sans mention")

Il n'y a pas d'argument « sinon » : on l'obtient en terminant par le test VRAI , toujours vrai, qui sert de cas par

défaut.

💡 ASTUCE

Disponibilité

SI.CONDITIONS existe à partir d'Excel 2019 et dans Microsoft 365. Dans Excel 2016 ou avant, elle affiche #NOM?  :

restez sur les SI imbriqués. Voir la fiche versions d'Excel.

04 Au-delà de quatre tranches : une table

Pour un barème de commission ou une grille tarifaire, une petite table est plus facile à maintenir qu'une longue

formule. On cherche alors la tranche avec RECHERCHEX en mode « valeur inférieure ».

Barème en E1:F4, formule en C2

A B C D E F

1 Commercial CA Taux Seuil Taux

2 Alice 23 000 € =RECHERCHEX(B2;E2:E4;F2:F4;;-1) 0 € 2 %

3 Bruno 8 000 € 2 % 10 000 € 4 %

4 Chloé 41 000 € 6 % 30 000 € 6 %

Changer un taux ou ajouter une tranche se fait dans la table, sans toucher aux formules.

05 Pièges à éviter

⚠️ PIÈGE

Le cas par défaut oublié

Avec SI.CONDITIONS, si aucun test n'est vrai, la formule renvoie #N/A . Terminez toujours par VRAI;"valeur par
défaut" .

⚠️ PIÈGE

Les parenthèses des SI imbriqués

Chaque SI ouvre une parenthèse qu'il faut refermer à la fin : trois SI, trois parenthèses fermantes. Excel colore les

paires de parenthèses pour aider à les repérer.

⚠️ PIÈGE

Les bornes des tranches

« De 10 à 20 » : 20 est-il inclus ? Choisissez entre >=  et >  en connaissant la règle. C'est la source d'erreur la

plus fréquente sur les barèmes.

01 Définition

EN UNE PHRASE

ET renvoie VRAI si toutes les conditions sont vraies. OU renvoie VRAI si au moins une condition est

vraie. On les utilise presque toujours à l'intérieur d'un SI.

02 Syntaxe

SYNTAXE

=ET(condition1; [condition2]; ...)
=OU(condition1; [condition2]; ...)
=NON(condition)

ET

Toutes les conditions doivent être vraies. Plus on en

ajoute, plus le test est exigeant.

OU

Une seule condition vraie suffit. Plus on en ajoute, plus

le test est souple.

03 Exemples

Prime si objectif atteint ET ancienneté d'au moins 1 an

A B C D

1 Commercial Objectif atteint Ancienneté Prime

2 Alice Oui 3 =SI(ET(B2="Oui";C2>=1);500;0)

3 Bruno Oui 0,5 0

4 Chloé Non 4 0

FICHE DE COURS GRATUITE

FORMULES EXCEL | LOGIQUE

ET, OU
Une prime pour les commerciaux qui dépassent l'objectif ET

qui ont plus d'un an d'ancienneté, une alerte si le stock est

nul OU si le produit est en rupture chez le fournisseur : ET et

OU réunissent plusieurs tests en un seul.

ALERTE SI LE STOCK EST VIDE OU SI LE FOURNISSEUR EST EN RUPTURE

=SI(OU(B2=0;C2="Rupture");"Commander";"")

UNE VALEUR COMPRISE ENTRE DEUX BORNES

=SI(ET(B2>=10;B2<=20);"Dans la cible";"Hors cible")

NON INVERSE UN TEST

=SI(NON(B2="Annulé");C2;0)

04 Pièges à éviter

⚠️ PIÈGE

Écrire 10<=B2<=20

Cette écriture mathématique ne fonctionne pas dans Excel : il calcule d'abord 10<=B2 , qui vaut VRAI ou FAUX,

puis compare ce résultat à 20. La formule renvoie toujours FAUX, sans message d'erreur. Écrivez

ET(B2>=10;B2<=20) .

⚠️ PIÈGE

ET et OU sur toute une plage

=ET(B2:B10>0)  teste chaque cellule et renvoie un seul résultat pour toute la plage. Pour un résultat par ligne, écri‐

vez la formule sur la première ligne avec B2  seul, puis recopiez-la.

⚠️ PIÈGE

Confondre ET et OU dans la phrase

« Les clients de Lille et de Roubaix » se traduit par OU(B2="Lille";B2="Roubaix")  : un client ne peut pas habiter

les deux villes à la fois. Reformulez la règle avant de choisir.

01 Définition

EN UNE PHRASE

SIERREUR renvoie le résultat d'une formule s'il est correct, et une valeur de remplacement si la for‐

mule renvoie une erreur, quelle qu'elle soit.

02 Syntaxe

SYNTAXE

=SIERREUR(valeur; valeur_si_erreur)
=SI.NON.DISP(valeur; valeur_si_na)

SI.NON.DISP fonctionne de la même façon mais ne réagit qu'à l'erreur #N/A, celle des fonctions de recherche.

Les autres erreurs restent visibles.

03 Exemples

Taux de transformation, avec des leads parfois à zéro

A B C D

1 Canal Leads Ventes Taux

2 Salon 40 6 =SIERREUR(C2/B2;"")

3 Site web 0 0

4 LinkedIn 25 3 12 %

FICHE DE COURS GRATUITE

FORMULES EXCEL | LOGIQUE

SIERREUR et SI.NON.DISP
Une colonne de RECHERCHEV pleine de #N/A, une division

par zéro qui casse le total : les messages d'erreur sont

parfois normaux, mais ils polluent un tableau. SIERREUR les

remplace par ce que vous voulez. À manier avec

discernement, car elle masque aussi les vraies erreurs.

UNE RECHERCHE QUI AFFICHE « INCONNU » AU LIEU DE #N/A

=SI.NON.DISP(RECHERCHEV(A2;Produits!A:C;3;FAUX);"Inconnu")

REMPLACER L'ERREUR PAR 0, POUR QUE LES TOTAUX FONCTIONNENT

=SIERREUR(C2/B2;0)

04 SIERREUR ou SI.NON.DISP ?

Erreur SIERREUR SI.NON.DISP

#N/A (valeur introuvable) Remplacée Remplacée

#DIV/0! (division par zéro) Remplacée Reste visible

#VALEUR!, #REF!, #NOM? Remplacées Restent visibles

Pour une recherche, préférez SI.NON.DISP : « introuvable » est une situation normale, alors qu'une faute de

frappe dans le nom d'une fonction ( #NOM? ) ou une colonne supprimée ( #REF! ) doit rester visible.

05 Pièges à éviter

⚠️ PIÈGE

Masquer une vraie erreur

=SIERREUR(RECHERCHEV(...);"")  affiche une cellule vide aussi bien quand le produit n'existe pas que quand la

formule est cassée. Le tableau a l'air propre, mais il est faux. Mettez la formule au point d'abord, ajoutez SIERREUR

ensuite.

⚠️ PIÈGE

Remplacer par "" puis faire des calculs

Un texte vide ""  n'est pas un nombre : =A2*2  sur une telle cellule renvoie #VALEUR! . SOMME et MOYENNE

l'ignorent, mais les opérations directes non. Si la colonne doit servir à des calculs, remplacez l'erreur par 0.

⚠️ PIÈGE

La version d'Excel

SIERREUR existe depuis Excel 2007 et SI.NON.DISP depuis Excel 2013. Pour RECHERCHEX, inutile de les ajouter :

son quatrième argument gère déjà le cas « introuvable ».

01 Définition

EN UNE PHRASE

SOMME.SI.ENS additionne les valeurs d'une plage pour les seules lignes qui respectent tous les cri‐

tères indiqués. SOMME.SI fait la même chose avec un seul critère.

02 Syntaxe

SYNTAXE

=SOMME.SI(plage; critère; [somme_plage])
=SOMME.SI.ENS(somme_plage; plage_critères1; critère1; [plage_critères2; critère2]; ...)

⚠️ PIÈGE

L'ordre des arguments n'est pas le même

Dans SOMME.SI, la plage à additionner est à la fin. Dans SOMME.SI.ENS, elle est au début. Le plus simple est de

n'utiliser que SOMME.SI.ENS, même avec un seul critère : une seule syntaxe à retenir.

FICHE DE COURS GRATUITE

FORMULES EXCEL | CONDITIONS

SOMME.SI et
SOMME.SI.ENS
Le total des ventes de la région Nord, le CA d'un commercial

sur le mois de mars, les dépenses supérieures à 500 € : dès

qu'un total dépend d'une condition, c'est SOMME.SI.ENS.

C'est l'une des fonctions les plus utiles d'Excel, et l'une des

plus demandées en entreprise.

03 Exemples

Les ventes du trimestre

A B C D

1 Date Région Commercial Montant

2 12/01/2026 Nord Alice 1 200 €

3 25/01/2026 Sud Bruno 800 €

4 03/02/2026 Nord Bruno 2 500 €

5 18/02/2026 Nord Alice 600 €

6 09/03/2026 Sud Alice 1 900 €

LE TOTAL DE LA RÉGION NORD : 4 300 €

=SOMME.SI.ENS(D2:D6;B2:B6;"Nord")

LE NORD, POUR ALICE SEULEMENT : 1 800 €

=SOMME.SI.ENS(D2:D6;B2:B6;"Nord";C2:C6;"Alice")

LES VENTES D'AU MOINS 1 000 € : 5 600 €

=SOMME.SI.ENS(D2:D6;D2:D6;">=1000")

04 Écrire les critères

Critère Écriture Ce qu'il retient

Texte exact "Nord" Les cellules égales à Nord, sans tenir compte des majuscules

Différent de "<>Annulé" Tout sauf Annulé

Comparaison ">500" Les nombres strictement supérieurs à 500

Contient "*nord*" Les textes qui contiennent nord

Commence par "P-1*" Les textes qui commencent par P-1

Cellule vide "" Les cellules vides

Cellule non vide "<>" Les cellules remplies

L'étoile *  remplace n'importe quelle suite de caractères, le point d'interrogation ?  remplace un seul caractère.

05 Un critère placé dans une cellule

Pour faire un tableau de synthèse, on écrit le critère dans une cellule et on y fait référence. Avec un opérateur, il
faut coller l'opérateur et la cellule avec & .

Synthèse par région, formule en G2 recopiée vers le bas

A B C D E F G

1 Date Région Commercial Montant Région Total

2 12/01/2026 Nord Alice 1 200 € Nord =SOMME.SI.ENS($D$2:$D$6;$B$2:$B$6;F2)

3 25/01/2026 Sud Bruno 800 € Sud 2 700 €

SEUIL DANS LA CELLULE H1

=SOMME.SI.ENS(D2:D6;D2:D6;">="&H1)

06 Critères sur des dates

LES VENTES DE FÉVRIER 2026 : 3 100 €

=SOMME.SI.ENS(D2:D6;A2:A6;">="&DATE(2026;2;1);A2:A6;"<"&DATE(2026;3;1))

On encadre la période par deux critères sur la même colonne : à partir du 1er février, et avant le 1er mars. Cela

fonctionne quel que soit le nombre de jours du mois.

07 Une condition OU

SOMME.SI.ENS combine ses critères avec ET. Pour additionner le Nord ou le Sud, on additionne deux

SOMME.SI.ENS :

LE NORD PLUS LE SUD

=SOMME.SI.ENS(D2:D6;B2:B6;"Nord")+SOMME.SI.ENS(D2:D6;B2:B6;"Sud")

08 Pièges à éviter

⚠️ PIÈGE

Des plages de tailles différentes

Toutes les plages d'un SOMME.SI.ENS doivent avoir le même nombre de lignes : D2:D6  avec B2:B7  renvoie

#VALEUR! . Avec des tableaux structurés, ce risque disparaît.

⚠️ PIÈGE

L'opérateur hors des guillemets

=SOMME.SI.ENS(D2:D6;D2:D6;>=H1)  est refusé. L'opérateur est du texte : ">="&H1 .

⚠️ PIÈGE

Des espaces invisibles

« Nord » et « Nord » suivi d'une espace sont deux textes différents, et la ligne est ignorée sans prévenir. Nettoyez

la colonne avec SUPPRESPACE  en cas de doute.

01 Définition

EN UNE PHRASE

NB.SI.ENS compte le nombre de lignes qui respectent tous les critères indiqués. NB.SI fait la même

chose avec un seul critère.

02 Syntaxe

SYNTAXE

=NB.SI(plage; critère)
=NB.SI.ENS(plage_critères1; critère1; [plage_critères2; critère2]; ...)

Les critères s'écrivent exactement comme pour SOMME.SI.ENS : "Lille" , ">=10" , "<>Annulé" , "*bio*" ...

03 Exemples

Suivi des commandes

A B C D

1 Commande Ville Statut Montant

2 C-501 Lille Livrée 320 €

3 C-502 Paris En cours 1 150 €

4 C-503 Lille Livrée 89 €

5 C-504 Lille Annulée 410 €

6 C-505 Paris Livrée 760 €

FICHE DE COURS GRATUITE

FORMULES EXCEL | CONDITIONS

NB.SI et NB.SI.ENS
Combien de commandes livrées, combien de clients à Lille,

combien d'absences en mars ? NB.SI.ENS répond à toutes

les questions qui commencent par « combien ». Elle sert

aussi à repérer les doublons dans une liste.

Version V1.0

Dernière mise à

jour
30/10/2026

Niveau
2 / 3,

Intermédiaire

Durée de lecture 7 min

Auteur Clément Delobel

Version en ligne
mesetudesdecas.fr/form
ules-excel/nb-si-ens.h
tml

LES COMMANDES LIVRÉES : 3

=NB.SI(C2:C6;"Livrée")

LES COMMANDES LIVRÉES À LILLE : 2

=NB.SI.ENS(B2:B6;"Lille";C2:C6;"Livrée")

LES COMMANDES DE PLUS DE 500 € : 2

=NB.SI(D2:D6;">500")

04 Les autres fonctions pour compter

Fonction Ce qu'elle compte

NB(B2:B50) Les cellules qui contiennent un nombre

NBVAL(B2:B50) Les cellules non vides, quel que soit leur contenu

NB.VIDE(B2:B50) Les cellules vides

NB.SI(B2:B50;"*") Les cellules qui contiennent du texte

05 Repérer les doublons

En comptant combien de fois chaque valeur apparaît dans la liste, on repère les doublons : tout résultat supérieur

à 1 en est un.

Contrôle des doublons d'e-mails, formule en B2 recopiée

A B

1 E-mail Doublon ?

2 alice@exemple.fr =SI(NB.SI($A$2:$A$5;A2)>1;"Doublon";"")

3 bruno@exemple.fr

4 alice@exemple.fr Doublon

5 chloe@exemple.fr

💡 ASTUCE

Garder la première occurrence

Avec =NB.SI($A$2:A2;A2) , la plage s'agrandit en descendant : la formule vaut 1 à la première apparition, 2 à la

deuxième... On peut alors filtrer sur les valeurs supérieures à 1 pour supprimer seulement les répétitions.

06 Pièges à éviter

⚠️ PIÈGE

Le joker ne compte pas les nombres

NB.SI(B2:B50;"*")  ne compte que le texte. Pour toutes les cellules remplies, utilisez NBVAL .

⚠️ PIÈGE

Un critère sur une cellule

NB.SI(D2:D6;">E1")  cherche le texte « >E1 ». Pour comparer à la valeur de E1, écrivez ">"&E1 .

⚠️ PIÈGE

Les longs numéros

NB.SI convertit en nombre les textes qui ressemblent à des nombres. Au-delà de 15 chiffres, comme pour un nu‐

méro de carte ou un IBAN sans lettres, deux numéros différents peuvent être comptés comme identiques. Pour ce

cas, utilisez =SOMMEPROD(--(A2:A100=A2)) .

01 Définition

EN UNE PHRASE

RECHERCHEV cherche une valeur dans la première colonne d'un tableau et renvoie la valeur située

sur la même ligne, dans la colonne dont on indique le numéro.

Le V signifie « verticale » : la recherche descend le long d'une colonne.

02 Syntaxe

SYNTAXE

=RECHERCHEV(valeur_cherchée; table_matrice; no_index_col; [valeur_proche])

Argument Rôle

valeur_cherchée Ce que l'on cherche, par exemple une référence produit

table_matrice Le tableau, dont la première colonne contient les valeurs cherchées

no_index_col Le numéro de la colonne à renvoyer, en comptant à partir de 1

valeur_proche FAUX pour une correspondance exacte, VRAI pour une valeur approchée

FICHE DE COURS GRATUITE

FORMULES EXCEL | RECHERCHE

RECHERCHEV
Retrouver le prix d'un produit à partir de sa référence, le

nom d'un client à partir de son numéro : RECHERCHEV est

la fonction de recherche la plus utilisée d'Excel. Elle a ses

limites, et RECHERCHEX la remplace avantageusement,

mais vous la croiserez dans presque tous les classeurs.

Version V1.0

Dernière mise à

jour
30/10/2026

Niveau
2 / 3,

Intermédiaire

Durée de lecture 9 min

Auteur Clément Delobel

Version en ligne
mesetudesdecas.fr/form
ules-excel/recherchev.
html

03 Exemple : retrouver un prix

Catalogue en A1:C6, recherche en F2

A B C D E F

1 Référence Produit Prix Réf. cherchée Prix

2 P-101 Clavier 49 € P-103 =RECHERCHEV(E2;$A$2:$C$6;3;FAUX)

3 P-102 Souris 25 €

4 P-103 Écran 189 €

5 P-104 Casque 79 €

6 P-105 Webcam 59 €

Excel descend la colonne A jusqu'à trouver P-103, en ligne 4, puis renvoie la 3e colonne de cette ligne : 189 €.

💡 ASTUCE

Toujours FAUX, sauf barème

Dans 95 % des cas, le dernier argument doit être FAUX  (ou 0 ). S'il est omis, Excel utilise VRAI  : il renvoie alors

une valeur approchée, souvent fausse, sans aucun message d'erreur.

04 Chercher dans une autre feuille

LE CATALOGUE EST SUR LA FEUILLE PRODUITS

=RECHERCHEV(A2;Produits!$A$2:$C$500;3;FAUX)

Tapez la formule jusqu'au premier point-virgule, puis cliquez sur l'onglet de la feuille et sélectionnez le tableau :

Excel écrit la référence. Pensez à F4  pour la bloquer avant de recopier.

05 La recherche approchée : les barèmes

Avec VRAI , RECHERCHEV renvoie la ligne de la plus grande valeur inférieure ou égale à la valeur cherchée. Cela
sert aux tranches et aux barèmes, à condition que la première colonne soit triée par ordre croissant.

Taux de remise selon le montant de la commande

A B C D E

1 Montant Remise À partir de Taux

2 2 450 € =RECHERCHEV(A2;$D$2:$E$5;2;VRAI) 0 € 0 %

3 780 € 5 % 500 € 5 %

4 6 100 € 15 % 1 000 € 10 %

5 5 000 € 15 %

06 Pièges à éviter

⚠️ PIÈGE

Chercher vers la gauche

RECHERCHEV ne regarde que la première colonne du tableau et ne renvoie que des colonnes situées à sa droite.

Pour retrouver une référence à partir d'un nom de produit placé à droite, utilisez RECHERCHEX ou INDEX + EQUIV.

⚠️ PIÈGE

Le numéro de colonne en dur

Si quelqu'un insère une colonne dans le tableau, le 3  de la formule pointe sur la mauvaise colonne, et le résultat

devient faux sans message d'erreur. RECHERCHEX n'a pas ce défaut.

⚠️ PIÈGE

Le tableau qui glisse

Sans $ , la plage A2:C6  devient A3:C7 , puis A4:C8  en recopiant la formule : les premières lignes du catalogue

sortent de la recherche et des #N/A  apparaissent. Bloquez-la avec F4 , ou utilisez un tableau structuré.

⚠️ PIÈGE

#N/A alors que la valeur existe

Le plus souvent, l'une des deux valeurs est un nombre et l'autre un nombre stocké en texte, ou l'une contient une

espace en trop. Voir la fiche #N/A pour les solutions.

01 Définition

EN UNE PHRASE

RECHERCHEX cherche une valeur dans une colonne (ou une ligne) et renvoie la valeur correspon‐

dante d'une autre colonne, où qu'elle se trouve.

💡 ASTUCE

Disponibilité

RECHERCHEX existe dans Excel 2021, Excel 2024, Microsoft 365 et Excel pour le web. Dans Excel 2019 ou avant,

elle affiche #NOM?  : utilisez RECHERCHEV ou INDEX + EQUIV.

02 Syntaxe

SYNTAXE

=RECHERCHEX(valeur_cherchée; tableau_recherche; tableau_renvoyé; [si_non_trouvé]; [mode_correspondance]; 
[mode_recherche])

Argument Rôle

valeur_cherchée Ce que l'on cherche

tableau_recherche La colonne où chercher

tableau_renvoyé La colonne d'où vient le résultat

si_non_trouvé Le texte ou la valeur à afficher si rien n'est trouvé. Sinon : #N/A

mode_correspondance 0 exacte (par défaut), -1 exacte ou valeur inférieure, 1 exacte ou valeur supérieure, 2 caractères

génériques

FICHE DE COURS GRATUITE

FORMULES EXCEL | RECHERCHE

RECHERCHEX
RECHERCHEX corrige tous les défauts de RECHERCHEV :

elle cherche dans n'importe quelle colonne, renvoie aussi

les colonnes de gauche, est exacte par défaut et gère elle-

même le cas « introuvable ». Si votre version d'Excel la

propose, c'est elle qu'il faut apprendre.

Version V1.0

Dernière mise à

jour
30/10/2026

Niveau
2 / 3,

Intermédiaire

Durée de lecture 9 min

Auteur Clément Delobel

Version en ligne
mesetudesdecas.fr/form
ules-excel/recherchex.
html

Argument Rôle

mode_recherche 1 du haut vers le bas (par défaut), -1 du bas vers le haut

03 L'exemple de base

Catalogue en A1:C5

A B C D E F

1 Référence Produit Prix Produit cherché Référence

2 P-101 Clavier 49 € Écran =RECHERCHEX(E2;B2:B5;A2:A5;"Inconnu")

3 P-102 Souris 25 €

4 P-103 Écran 189 €

5 P-104 Casque 79 €

On cherche le nom du produit en colonne B et on renvoie la référence en colonne A, à gauche : impossible avec

RECHERCHEV. Et si le produit n'existe pas, la cellule affiche « Inconnu » au lieu de #N/A .

04 Ce qui change par rapport à RECHERCHEV

RECHERCHEV RECHERCHEX

Cherche seulement dans la première colonne Cherche dans n'importe quelle colonne

Renvoie seulement vers la droite Renvoie dans toutes les directions

Numéro de colonne en dur Colonne désignée directement : insérer une colonne ne casse rien

Approchée par défaut si on oublie FAUX Exacte par défaut

Introuvable : #N/A, à masquer avec SIERREUR Argument si_non_trouvé intégré

Renvoie la première occurrence Première ou dernière occurrence au choix

05 Plus loin avec RECHERCHEX

RENVOYER PLUSIEURS COLONNES D'UN COUP : PRODUIT ET PRIX

=RECHERCHEX(E2;A2:A6;B2:C6)

Le résultat déborde sur deux cellules côte à côte : c'est un tableau dynamique.

LA DERNIÈRE COMMANDE D'UN CLIENT, EN CHERCHANT DU BAS VERS LE HAUT

=RECHERCHEX(E2;B2:B500;D2:D500;"Aucune";0;-1)

UNE RECHERCHE SUR DEUX CRITÈRES : CLIENT ET MOIS

=RECHERCHEX(F2&G2;A2:A500&B2:B500;C2:C500;"Introuvable")

UN BARÈME : LE TAUX DE LA TRANCHE INFÉRIEURE (PAS BESOIN DE TRIER)

=RECHERCHEX(B2;$E$2:$E$5;$F$2:$F$5;;-1)

06 Pièges à éviter

⚠️ PIÈGE

Des plages de tailles différentes

tableau_recherche  et tableau_renvoyé  doivent avoir le même nombre de lignes : A2:A6  avec C2:C7  renvoie

#VALEUR! .

⚠️ PIÈGE

Le fichier partagé

Un collègue qui ouvre le fichier avec Excel 2016 ou 2019 verra #NOM?  dans toutes les cellules. Avant d'envoyer un

classeur, vérifiez la version de vos destinataires.

⚠️ PIÈGE

Un débordement bloqué

Si RECHERCHEX renvoie plusieurs colonnes et que les cellules voisines ne sont pas vides, la formule affiche

#PROPAGATION! . Videz les cellules qui gênent.

01 Définition

EN UNE PHRASE

EQUIV trouve la position d'une valeur dans une liste. INDEX renvoie la valeur située à une position

donnée. Ensemble, ils font une recherche : EQUIV trouve la ligne, INDEX va chercher le résultat.

02 Syntaxe

SYNTAXE

=EQUIV(valeur_cherchée; tableau_recherche; [type])
=INDEX(matrice; no_lig; [no_col])

Pour EQUIV, le type 0  donne une correspondance exacte. C'est presque toujours celui qu'il faut.

03 Chaque fonction séparément

Les deux fonctions seules

A B C D E

1 Produit Prix Formule Résultat

2 Clavier 49 € =EQUIV("Écran";A2:A5;0) 3

3 Souris 25 € =INDEX(B2:B5;3) 189 €

4 Écran 189 €

5 Casque 79 €

EQUIV répond « Écran est en 3e position ». INDEX répond « la 3e valeur de la colonne des prix est 189 € ».

FICHE DE COURS GRATUITE

FORMULES EXCEL | RECHERCHE

INDEX + EQUIV
Avant RECHERCHEX, les utilisateurs avancés combinaient

deux fonctions pour contourner les limites de RECHERCHEV.

Ce duo reste précieux : il fonctionne dans toutes les

versions d'Excel et il gère les recherches à double entrée,

ligne et colonne.

Version V1.0

Dernière mise à

jour
30/10/2026

Niveau 3 / 3, Avancé

Durée de lecture 9 min

Auteur Clément Delobel

Version en ligne
mesetudesdecas.fr/form
ules-excel/index-equi
v.html

04 Les deux ensemble

LE PRIX DU PRODUIT ÉCRIT EN E2

=INDEX(B2:B5;EQUIV(E2;A2:A5;0))

La colonne cherchée et la colonne renvoyée sont indépendantes : elles peuvent être dans n'importe quel ordre, y

compris le résultat à gauche.

05 La recherche à double entrée

Avec deux EQUIV, l'un pour la ligne et l'autre pour la colonne, INDEX lit la valeur au croisement d'un tableau.

Tarif selon la zone (ligne) et le poids (colonne)

A B C D

1 Zone 1 kg 5 kg 10 kg

2 France 7 € 12 € 19 €

3 Europe 15 € 24 € 38 €

4 Monde 28 € 45 € 72 €

ZONE EN G1 (EUROPE), POIDS EN G2 (5 KG) : 24 €

=INDEX(B2:D4;EQUIV(G1;A2:A4;0);EQUIV(G2;B1:D1;0))

06 Pièges à éviter

⚠️ PIÈGE

Oublier le 0 d'EQUIV

Sans le troisième argument, EQUIV cherche une valeur approchée et suppose la liste triée. Sur une liste non triée, il

renvoie une position fausse sans message d'erreur. Écrivez toujours ;0 .

⚠️ PIÈGE

Deux plages décalées

INDEX(B3:B6;EQUIV(E2;A2:A5;0))  : les plages ne commencent pas à la même ligne, tous les résultats sont déca‐

lés d'une ligne. Vérifiez que les deux plages ont la même hauteur et le même départ.

⚠️ PIÈGE

Quand préférer RECHERCHEX

Si tous les utilisateurs ont Excel 2021 ou plus, RECHERCHEX fait la même chose en une seule fonction, plus lisible.

INDEX + EQUIV reste le bon choix pour les fichiers partagés avec d'anciennes versions.

01 Définition

EN UNE PHRASE

Concaténer, c'est mettre bout à bout plusieurs textes ou valeurs dans une seule cellule. L'opérateur

&  suffit dans la plupart des cas.

02 L'opérateur &

Prénom et nom réunis

A B C

1 Prénom Nom Nom complet

2 Alice Martin =A2&" "&B2

3 Bruno Leroy Bruno Leroy

Le " "  ajoute l'espace entre les deux : sans lui, on obtiendrait « AliceMartin ».

UNE PHRASE COMPLÈTE

="Commande "&A2&" livrée à "&B2

03 CONCAT et JOINDRE.TEXTE

SYNTAXE

=CONCAT(texte1; [texte2]; ...)
=JOINDRE.TEXTE(délimiteur; ignorer_vide; texte1; [texte2]; ...)

FICHE DE COURS GRATUITE

FORMULES EXCEL | TEXTE

Assembler du texte

Prénom et nom dans une seule cellule, une phrase qui

reprend un montant, un identifiant fabriqué à partir de

plusieurs colonnes : assembler du texte est un besoin de

tous les jours. Le plus simple est l'esperluette, mais elle a un

piège avec les dates et les nombres.

Version V1.0

Dernière mise à

jour
30/10/2026

Niveau 1 / 3, Découverte

Durée de lecture 7 min

Auteur Clément Delobel

Version en ligne
mesetudesdecas.fr/form
ules-excel/concatener.
html

CONCAT accepte des plages entières : =CONCAT(A2:D2)  colle les quatre cellules. JOINDRE.TEXTE ajoute en plus

un séparateur entre chaque valeur, et peut ignorer les cellules vides :

Adresse complète, en sautant le complément vide

A B C D E

1 Rue Complément CP Ville Adresse

2 12 rue Nationale 59000 Lille =JOINDRE.TEXTE(", ";VRAI;A2:D2)

Résultat : « 12 rue Nationale, 59000, Lille », sans virgule en double.

💡 ASTUCE

Disponibilité

CONCAT et JOINDRE.TEXTE existent à partir d'Excel 2019. L'ancienne fonction CONCATENER fonctionne encore,

mais n'accepte pas les plages.

04 Garder le format des nombres et des dates

Une date ou un montant perd sa mise en forme dès qu'on l'assemble avec du texte, car Excel utilise la valeur

brute. La fonction TEXTE applique un format choisi :

Formule Résultat

="Échéance : "&B2 Échéance : 46325

="Échéance : "&TEXTE(B2;"jj/mm/aaaa") Échéance : 30/10/2026

="Total : "&TEXTE(C2;"# ##0,00 €") Total : 1 250,00 €

=TEXTE(B2;"mmmm aaaa") octobre 2026

05 Un retour à la ligne dans la cellule

NOM SUR UNE LIGNE, VILLE SUR LA SUIVANTE

=A2&CAR(10)&B2

CAR(10)  est le caractère de saut de ligne. Il ne s'affiche que si la cellule a l'option Renvoyer à la ligne automati‐

quement (onglet Accueil).

06 Pièges à éviter

⚠️ PIÈGE

Le résultat est du texte

="1"&"5"  donne le texte « 15 », pas le nombre 15 : il ne s'additionne pas. Si vous fabriquez un nombre, convertis‐

sez-le avec CNUM , ou multipliez-le par 1.

⚠️ PIÈGE

Les guillemets dans le texte

Pour afficher un guillemet, il faut le doubler : ="Il a dit ""oui"""  affiche Il a dit "oui".

⚠️ PIÈGE

Assembler sans figer

Le résultat d'une concaténation reste une formule : si vous supprimez les colonnes d'origine, il disparaît. Pour gar‐

der seulement le texte, copiez puis Collage spécial > Valeurs.

01 Définition

EN UNE PHRASE

GAUCHE renvoie les premiers caractères d'un texte, DROITE les derniers, et STXT ceux du milieu, à

partir d'une position donnée.

02 Syntaxe

SYNTAXE

=GAUCHE(texte; [nb_car])
=DROITE(texte; [nb_car])
=STXT(texte; no_départ; nb_car)
=NBCAR(texte)
=TROUVE(texte_cherché; texte; [no_départ])

NBCAR compte les caractères d'un texte. TROUVE donne la position d'un caractère : =TROUVE(" ";"Alice Mar-
tin")  renvoie 6.

03 Longueur fixe

Quand l'information a toujours la même longueur, les trois fonctions suffisent.

Découper un numéro de dossier

A B C D

1 Dossier Année Service Numéro

2 2026-RH-0457 =GAUCHE(A2;4) =STXT(A2;6;2) =DROITE(A2;4)

3 2025-CO-1203 2025 CO 1203

FICHE DE COURS GRATUITE

FORMULES EXCEL | TEXTE

GAUCHE, DROITE, STXT
Isoler le code postal d'une adresse, le prénom d'un nom

complet, l'année d'un numéro de dossier : les exports des

logiciels mettent souvent plusieurs informations dans la

même cellule. Trois fonctions découpent le texte, deux

autres trouvent où couper.

Version V1.0

Dernière mise à

jour

Niveau

Intermédiaire

Durée de lecture 8 min

04 Couper avant ou après un séparateur

Quand la longueur varie, on cherche d'abord la position du séparateur avec TROUVE.

Prénom et nom séparés par une espace

A B C

1 Nom complet Prénom Nom

2 Alice Martin =GAUCHE(A2;TROUVE(" ";A2)-1) =STXT(A2;TROUVE(" ";A2)+1;100)

3 Jean-Pierre Durand Jean-Pierre Durand

Le -1  retire l'espace du prénom. Pour le nom, on part juste après l'espace et on prend 100 caractères : STXT

s'arrête de lui-même à la fin du texte.

💡 ASTUCE

Plus simple dans Excel récent

Dans Excel 2024 et Microsoft 365, =TEXTE.AVANT(A2;" ")  et =TEXTE.APRES(A2;" ")  font la même chose sans

calcul de position. Voir aussi les nouvelles fonctions.

05 Sans formule : le remplissage instantané

Tapez le résultat attendu pour la première ligne, par exemple « Alice » à côté de « Alice Martin », puis appuyez sur

Ctrl + E  dans la cellule du dessous. Excel devine la règle et remplit toute la colonne. Pratique pour un net‐

toyage ponctuel, mais le résultat ne se met pas à jour si les données changent.

06 Pièges à éviter

⚠️ PIÈGE

Le résultat est toujours du texte

=GAUCHE("59000 Lille";5)  renvoie le texte « 59000 », pas un nombre. Pour calculer avec, ou pour le comparer à

un nombre dans une recherche, convertissez-le : =CNUM(GAUCHE(A2;5)) .

⚠️ PIÈGE

Séparateur absent : #VALEUR!

Si une cellule ne contient pas d'espace, TROUVE renvoie #VALEUR! . Protégez la formule avec SIERREUR, par

exemple =SIERREUR(GAUCHE(A2;TROUVE(" ";A2)-1);A2) .

⚠️ PIÈGE

Les espaces en trop

Un export peut contenir des espaces au début, à la fin ou en double, qui faussent les positions. Commencez par

nettoyer avec =SUPPRESPACE(A2) .

⚠️ PIÈGE

TROUVE ou CHERCHE

TROUVE respecte les majuscules et minuscules, CHERCHE non et accepte les jokers *  et ? . Pour un séparateur

comme l'espace ou le tiret, les deux donnent le même résultat.

01 Définition

EN UNE PHRASE

DATEDIF calcule l'écart entre deux dates en années, en mois ou en jours entiers, selon l'unité

demandée.

02 Syntaxe

SYNTAXE

=DATEDIF(date_début; date_fin; unité)

Code Ce qu'il renvoie

"y" Le nombre d'années complètes

"m" Le nombre de mois complets

"d" Le nombre de jours

"ym" Les mois restants, une fois les années complètes retirées

"yd" Les jours restants, une fois les années complètes retirées

Les codes s'écrivent en anglais, entre guillemets, même dans un Excel en français.

FICHE DE COURS GRATUITE

FORMULES EXCEL | DATES

DATEDIF
Calculer l'âge d'un salarié ou son ancienneté paraît simple,

jusqu'à ce qu'on essaie de diviser des jours par 365.

DATEDIF fait le calcul exact en années, en mois ou en jours.

C'est une fonction cachée : elle n'apparaît ni dans l'assistant

ni dans la saisie semi-automatique, mais elle fonctionne

dans toutes les versions.

Version V1.0

Dernière mise à

jour

Niveau

03 Exemples

Âge et ancienneté au 30/10/2026 (cellule C1 pour la date de calcul)

A B C D E

1 Salarié Naissance Entrée Âge Ancienneté

2 Alice 14/03/1988 02/09/2019 =DATEDIF(B2;$C$1;"y") =DATEDIF(C2;$C$1;"y")&" ans "&DATEDIF(C2;$C$1;"ym")&

Alice a 38 ans, et 7 ans 1 mois d'ancienneté. Pour calculer à la date du jour, remplacez la cellule de date par AU-
JOURDHUI()  :

L'ÂGE À LA DATE DU JOUR

=DATEDIF(B2;AUJOURDHUI();"y")

💡 ASTUCE

Juste un nombre de jours

Pour un simple écart en jours, une soustraction suffit : =C2-B2 . Une date est un nombre pour Excel, et la diffé‐

rence de deux dates est un nombre de jours. Si le résultat s'affiche comme une date, appliquez le format Standard.

04 Pièges à éviter

⚠️ PIÈGE

Date de début après la date de fin

DATEDIF renvoie #NOMBRE!  si la première date est postérieure à la seconde. Vérifiez l'ordre des arguments.

⚠️ PIÈGE

Le code "md"

Il existe un code "md"  (jours restants une fois les mois retirés), mais Microsoft déconseille de l'utiliser : il peut ren‐

voyer un résultat faux, voire négatif, selon les dates.

⚠️ PIÈGE

Diviser par 365

=(C2-B2)/365  donne un âge approximatif, faux autour de l'anniversaire à cause des années bissextiles. DATEDIF

avec "y"  donne l'âge exact.

⚠️ PIÈGE

Une date saisie comme du texte

Une date importée peut être du texte : elle est alignée à gauche. DATEDIF renvoie alors #VALEUR! . Voir la fiche

#VALEUR!.

01 Définition

EN UNE PHRASE

NB.JOURS.OUVRES compte les jours du lundi au vendredi entre deux dates, bornes comprises, en

retirant les jours fériés qu'on lui indique.

02 Syntaxe

SYNTAXE

=NB.JOURS.OUVRES(date_départ; date_fin; [jours_fériés])
=NB.JOURS.OUVRES.INTL(date_début; date_fin; [week-end]; [jours_fériés])
=SERIE.JOUR.OUVRE(date_départ; nb_jours; [jours_fériés])

03 Préparer la liste des jours fériés

Excel ne connaît pas les jours fériés : il faut les lister dans une plage, idéalement sur une feuille à part, et lui don‐

ner un nom, par exemple Feries .

FICHE DE COURS GRATUITE

FORMULES EXCEL | DATES

NB.JOURS.OUVRES
Combien de jours de travail dans ce projet, quelle date de

livraison à 10 jours ouvrés ? Les week-ends et les jours

fériés rendent le calcul pénible à la main. Trois fonctions

s'en chargent, à condition de leur donner la liste des jours

fériés.

Version V1.0

Dernière mise à

jour

Durée de lecture 7 min

Jours fériés 2026 en France métropolitaine (plage nommée Feries)

A B

1 Date Jour férié

2 01/01/2026 Jour de l'an

3 06/04/2026 Lundi de Pâques

4 01/05/2026 Fête du travail

5 08/05/2026 Victoire 1945

6 14/05/2026 Ascension

7 25/05/2026 Lundi de Pentecôte

8 14/07/2026 Fête nationale

9 15/08/2026 Assomption

10 01/11/2026 Toussaint

11 11/11/2026 Armistice

12 25/12/2026 Noël

04 Exemples

JOURS OUVRÉS EN NOVEMBRE 2026 : 20

=NB.JOURS.OUVRES(DATE(2026;11;1);DATE(2026;11;30);Feries)

Novembre 2026 compte 21 jours du lundi au vendredi. Le 11 novembre tombe un mercredi : il est retiré. Le 1er no‐

vembre tombe un dimanche : il n'était de toute façon pas compté.

LA DATE DE LIVRAISON, 10 JOURS OUVRÉS APRÈS LE 30/04/2026 : 19/05/2026

=SERIE.JOUR.OUVRE(DATE(2026;4;30);10;Feries)

Le résultat s'affiche parfois comme un nombre : appliquez le format Date courte.

05 D'autres week-ends : NB.JOURS.OUVRES.INTL

Pour un commerce fermé seulement le dimanche, ou une équipe qui ne travaille pas le lundi, l'argument week-

end précise les jours non travaillés.

Code Jours non travaillés

1 Samedi et dimanche (par défaut)

11 Dimanche seulement

"0000011" Une chaîne de 7 chiffres, du lundi au dimanche : 1 = non travaillé

"1000001" Lundi et dimanche

JOURS D'OUVERTURE D'UN MAGASIN OUVERT DU LUNDI AU SAMEDI

=NB.JOURS.OUVRES.INTL(B2;C2;11;Feries)

06 Pièges à éviter

⚠️ PIÈGE

Les deux bornes sont comptées

Du lundi au vendredi de la même semaine, NB.JOURS.OUVRES renvoie 5, pas 4. Pour un délai « à partir du lende‐

main », retirez 1.

⚠️ PIÈGE

Une liste de fériés à mettre à jour

Pâques, l'Ascension et la Pentecôte changent chaque année. Une liste 2026 ne vaut pas pour 2027 : prévoyez une

ligne par année, ou complétez la liste chaque début d'année.

⚠️ PIÈGE

Jours ouvrés ou ouvrables

En droit du travail, les jours ouvrables vont du lundi au samedi, les jours ouvrés du lundi au vendredi. Pour des

jours ouvrables, utilisez NB.JOURS.OUVRES.INTL avec le code 11 .

01 Définition

EN UNE PHRASE

FILTRE renvoie toutes les lignes d'un tableau qui respectent une condition. Le résultat occupe au‐

tant de cellules que nécessaire : il « déborde » de la cellule de la formule.

02 Syntaxe

SYNTAXE

=FILTRE(tableau; inclure; [si_vide])

Argument Rôle

tableau Les données à renvoyer, une ou plusieurs colonnes

inclure Le test, ligne par ligne, par exemple B2:B20="Nord"

si_vide Ce qu'il faut afficher si aucune ligne ne correspond

FICHE DE COURS GRATUITE

FORMULES EXCEL | TABLEAUX DYNAMIQUES

La fonction FILTRE
Le filtre automatique masque des lignes, la fonction FILTRE

les recopie ailleurs, et se met à jour toute seule quand les

données changent. Un rapport « commandes en retard » ou

« ventes du Nord » se construit en une seule formule.

Version V1.0

jour

Niveau

03 Exemple

Ventes en A1:D6, formule en F2

A B C D E F

1 Date Région Commercial Montant Résultat

2 12/01/2026 Nord Alice 1 200 € =FILTRE(A2:D6;B2:B6="Nord";"Aucune vente")

3 25/01/2026 Sud Bruno 800 €

4 03/02/2026 Nord Bruno 2 500 €

5 18/02/2026 Nord Alice 600 €

6 09/03/2026 Sud Alice 1 900 €

La formule, écrite une seule fois en F2, remplit F2:I4 avec les trois ventes du Nord. Un cadre bleu entoure la zone

de débordement quand on clique dedans.

04 Plusieurs critères

Dans l'argument inclure, on combine les tests avec deux opérateurs :

*  pour ET : toutes les conditions doivent être vraies ;

+  pour OU : une seule suffit.

Chaque test est mis entre parenthèses.

LE NORD ET PLUS DE 1 000 €

=FILTRE(A2:D6;(B2:B6="Nord")*(D2:D6>1000))

ALICE OU BRUNO

=FILTRE(A2:D6;(C2:C6="Alice")+(C2:C6="Bruno"))

CRITÈRE PLACÉ DANS UNE CELLULE

=FILTRE(A2:D6;B2:B6=H1;"Aucune vente")

05 Trier le résultat

LES VENTES DU NORD, DE LA PLUS GROSSE À LA PLUS PETITE

=TRIER(FILTRE(A2:D6;B2:B6="Nord");4;-1)

TRIER prend le résultat de FILTRE, le trie sur sa 4e colonne (le montant), en ordre décroissant ( -1 ).

💡 ASTUCE

Disponibilité

FILTRE et TRIER existent dans Excel 2021, Excel 2024, Microsoft 365 et Excel pour le web. Voir la fiche versions

d'Excel.

06 Pièges à éviter

⚠️ PIÈGE

#CALC! quand rien ne correspond

Sans troisième argument, une recherche sans résultat renvoie #CALC! . Ajoutez toujours un message : "Aucune

vente" .

⚠️ PIÈGE

#PROPAGATION!

Le résultat a besoin de place. Si une cellule de la zone de débordement contient déjà quelque chose, la formule af‐

fiche #PROPAGATION! . Videz la zone, ou déplacez la formule.

⚠️ PIÈGE

Écrire ET au lieu de *

=FILTRE(A2:D6;ET(B2:B6="Nord";D2:D6>1000))  ne fonctionne pas : ET renvoie un seul résultat pour toute la

plage, pas un résultat par ligne. Utilisez la multiplication.

⚠️ PIÈGE

Des plages de hauteurs différentes

Le tableau et le test doivent avoir le même nombre de lignes : A2:D6  avec B2:B7  renvoie #VALEUR! .

mesétudesdecas.fr
Formules Excel | Recueil complet

V1.0 - 30/10/2026

FALE | Formation, Accompagnement, Logiciels, Expertise | EURL au capital de 2 000 € | RCS Lille Métropole n° 933 672 537

229 rue de Solférino, 59000 Lille | TVA FR33 933 672 537 | c.delobel@fale.fr | Toutes les fiches gratuites : mesetudesdecas.fr/formules-excel |
© FALE

Page 51 / 66



01 Définition

EN UNE PHRASE

UNIQUE renvoie la liste des valeurs différentes d'une plage, chacune une seule fois, dans l'ordre de

leur première apparition.

02 Syntaxe

SYNTAXE

=UNIQUE(tableau; [par_col]; [exactement_une_fois])

Le troisième argument, mis à VRAI , ne garde que les valeurs qui apparaissent une seule fois dans la liste.

03 Exemple

Liste des commerciaux sans doublons, formule en F2

A B C D E F

1 Date Région Commercial Montant Commerciaux

2 12/01/2026 Nord Alice 1 200 € =UNIQUE(C2:C6)

3 25/01/2026 Sud Bruno 800 € Bruno

4 03/02/2026 Nord Bruno 2 500 €

5 18/02/2026 Nord Alice 600 €

6 09/03/2026 Sud Alice 1 900 €

FICHE DE COURS GRATUITE

FORMULES EXCEL | TABLEAUX DYNAMIQUES

La fonction UNIQUE
La liste des clients, des régions ou des produits d'un

tableau, sans doublons et toujours à jour : avant, il fallait un

tableau croisé ou l'outil Supprimer les doublons, à refaire à

chaque ajout. La fonction UNIQUE le fait en une formule.

Version V1.0

Auteur Clément Delobel

Version en ligne
mesetudesdecas.fr/form
ules-excel/unique.html

mesétudesdecas.fr
Formules Excel | Recueil complet

V1.0 - 30/10/2026

FALE | Formation, Accompagnement, Logiciels, Expertise | EURL au capital de 2 000 € | RCS Lille Métropole n° 933 672 537

229 rue de Solférino, 59000 Lille | TVA FR33 933 672 537 | c.delobel@fale.fr | Toutes les fiches gratuites : mesetudesdecas.fr/formules-excel |
© FALE

Page 52 / 66



04 Avec d'autres fonctions

LA LISTE TRIÉE PAR ORDRE ALPHABÉTIQUE

=TRIER(UNIQUE(C2:C6))

LE NOMBRE DE COMMERCIAUX DIFFÉRENTS : 2

=NBVAL(UNIQUE(C2:C6))

LES COUPLES RÉGION ET COMMERCIAL, SANS DOUBLONS

=UNIQUE(B2:C6)

05 Un tableau de synthèse qui s'agrandit seul

La zone de débordement d'une formule se désigne avec le signe #  : F2#  veut dire « toute la liste qui part de F2

». On peut ainsi construire une synthèse dont le nombre de lignes suit les données.

EN F2, LA LISTE ; EN G2, LE TOTAL DE CHAQUE NOM DE LA LISTE

=UNIQUE(C2:C6)
=SOMME.SI.ENS(D2:D6;C2:C6;F2#)

La formule en G2 déborde elle aussi : un total par nom de la liste. Ajoutez un commercial dans les données, une

ligne apparaît dans la synthèse.

💡 ASTUCE

Une liste déroulante qui suit les données

Dans Données > Validation des données > Liste, indiquez la source =$F$2#  : la liste déroulante contient toujours

les valeurs à jour, sans doublons.

06 Pièges à éviter

⚠️ PIÈGE

Les cellules vides

Si la plage contient des cellules vides, UNIQUE renvoie une ligne avec 0. Excluez-les avec FILTRE :

=UNIQUE(FILTRE(C2:C100;C2:C100<>"")) .

⚠️ PIÈGE

Les variantes d'écriture

« Alice », « alice » et « Alice » suivi d'une espace : UNIQUE considère les deux premières comme identiques, mais

la troisième comme différente. Nettoyez la colonne avec SUPPRESPACE  avant.

⚠️ PIÈGE

La version d'Excel

UNIQUE existe dans Excel 2021, Excel 2024 et Microsoft 365. Dans les versions précédentes, utilisez Données >

Supprimer les doublons sur une copie, ou un tableau croisé dynamique.

01 Définition

EN UNE PHRASE

Depuis Excel 2021, les formules peuvent renvoyer plusieurs valeurs qui débordent sur les cellules

voisines. Cette évolution a permis une nouvelle génération de fonctions : filtrer, trier, découper du

texte, regrouper des données, sans macro ni tableau croisé.

02 Panorama des fonctions récentes

Fonction Rôle Disponible dans

FILTRE, TRIER, UNIQUE Extraire, trier, dédoublonner Excel 2021 et suivants

RECHERCHEX, EQUIVX Rechercher sans les limites de

RECHERCHEV

Excel 2021 et suivants

SEQUENCE, LET Générer une suite, nommer un calcul

intermédiaire

Excel 2021 et suivants

TEXTE.AVANT, TEXTE.APRES, FRACTIONNER.TEXTE Découper un texte Excel 2024 et Microsoft

365

ASSEMB.V, ASSEMB.H, CHOISIRCOLS, CHOISIRLIGNES Empiler et remodeler des tableaux Excel 2024 et Microsoft

365

LAMBDA Créer ses propres fonctions, sans VBA Excel 2024 et Microsoft

365

REGEXTEST, REGEXEXTRACT, REGEXREPLACE Expressions régulières Microsoft 365

uniquement

GROUPER.PAR, PIVOTER.PAR Tableaux croisés en formule Microsoft 365

uniquement

FICHE DE COURS GRATUITE

FORMULES EXCEL | VERSIONS

Les nouvelles
fonctions d'Excel
Excel a plus changé depuis 2020 qu'au cours des quinze

années précédentes. Les dernières fonctions savent

analyser du texte avec des expressions régulières et

produire un tableau croisé dynamique en une formule. Cette

fiche présente les plus utiles, et rappelle une ancienne

fonction encore méconnue : AGREGAT.

Version V1.0

Dernière mise à

jour
30/10/2026

Niveau 3 / 3, Avancé

Durée de lecture 10 min

Auteur Clément Delobel

Version en ligne
mesetudesdecas.fr/form
ules-excel/nouvelles-f
onctions.html

03 Les fonctions REGEX

Une expression régulière décrit un motif de texte : « 5 chiffres », « un e-mail », « tout ce qui est entre paren‐

thèses ». Trois fonctions l'utilisent :

SYNTAXE

=REGEXTEST(texte; modèle; [respect_casse])
=REGEXEXTRACT(texte; modèle; [mode_renvoi]; [respect_casse])
=REGEXREPLACE(texte; modèle; remplacement; [occurrence]; [respect_casse])

Motif Signification

\d Un chiffre

\d{5} Exactement cinq chiffres

[A-Z] Une lettre majuscule

\s+ Une ou plusieurs espaces

^ et $ Le début et la fin du texte

.+ N'importe quels caractères, au moins un

Extraire le code postal d'une adresse

A B

1 Adresse Code postal

2 12 rue Nationale 59000 Lille =REGEXEXTRACT(A2;"\d{5}")

3 3 place Bellecour, 69002 Lyon 69002

VÉRIFIER QU'UNE CELLULE RESSEMBLE À UN E-MAIL : VRAI OU FAUX

=REGEXTEST(A2;"^[^@\s]+@[^@\s]+\.[a-z]{2,}$")

REMPLACER LES ESPACES MULTIPLES PAR UNE SEULE ESPACE

=REGEXREPLACE(A2;"\s+";" ")

GARDER UNIQUEMENT LES CHIFFRES D'UN NUMÉRO DE TÉLÉPHONE

=REGEXREPLACE(A2;"\D";"")

💡 ASTUCE

Pas besoin de tout connaître

Quatre ou cinq motifs couvrent l'essentiel des besoins. Pour un motif plus complexe, décrivez ce que vous voulez à

un assistant IA, puis testez le résultat sur quelques lignes avant de recopier la formule.

04 GROUPER.PAR et PIVOTER.PAR

Ces deux fonctions produisent en une formule ce que fait un tableau croisé dynamique : regrouper et totaliser. Le
résultat se met à jour dès que les données changent, sans bouton Actualiser.

SYNTAXE SIMPLIFIÉE

=GROUPER.PAR(champs_lignes; valeurs; fonction)
=PIVOTER.PAR(champs_lignes; champs_colonnes; valeurs; fonction)

Ventes en A1:D6, total par région en F2

A B C D E F

1 Date Région Commercial Montant Résultat

2 12/01/2026 Nord Alice 1 200 € =GROUPER.PAR(B2:B6;D2:D6;SOMME)

3 25/01/2026 Sud Bruno 800 € Sud

4 03/02/2026 Nord Bruno 2 500 € Total

5 18/02/2026 Nord Alice 600 €

6 09/03/2026 Sud Alice 1 900 €

La formule renvoie Nord : 4 300 €, Sud : 2 700 €, puis une ligne de total général à 7 000 €.

RÉGIONS EN LIGNES, COMMERCIAUX EN COLONNES

=PIVOTER.PAR(B2:B6;C2:C6;D2:D6;SOMME)

Le troisième argument est le calcul à faire : SOMME, MOYENNE, NB, MAX... sans parenthèses. Les arguments fa‐

cultatifs suivants gèrent les en-têtes, les sous-totaux, le tri et un filtre.

⚠️ PIÈGE

Microsoft 365 uniquement

Les fonctions REGEX, GROUPER.PAR et PIVOTER.PAR n'existent pas dans Excel 2024 ni dans les versions précé‐

dentes, qui afficheront #NOM? . Réservez-les aux fichiers utilisés par des personnes équipées de Microsoft 365, ou

d'Excel pour le web.

05 AGREGAT : l'ancienne fonction à connaître

AGREGAT existe depuis Excel 2010, mais elle reste méconnue. Elle fait une somme, une moyenne, un maximum...

en ignorant les erreurs et les lignes masquées, là où SOMME renvoie une erreur dès qu'une cellule en contient

une.

SYNTAXE

=AGREGAT(no_fonction; options; réf1; ...)

Code Fonction Code Option

1 MOYENNE 5 Ignorer les lignes masquées

Code Fonction Code Option

4 MAX 6 Ignorer les erreurs

5 MIN 7 Ignorer les lignes masquées et les erreurs

9 SOMME 4 Ne rien ignorer

Un total malgré une erreur dans la colonne

A B

1 Produit Marge

2 Clavier 12 €

3 Souris #DIV/0!

4 Écran 41 €

5 Total =AGREGAT(9;6;B2:B4)

Résultat : 53 €. Avec =SOMME(B2:B4) , on obtiendrait #DIV/0! .

06 Pièges à éviter

⚠️ PIÈGE

Partager un fichier avec des fonctions récentes

Une cellule qui utilise une fonction absente de la version du destinataire affiche #NOM? . Avant d'envoyer un clas‐

seur, consultez la fiche versions d'Excel.

⚠️ PIÈGE

Les antislashs des expressions régulières

Dans un motif, \d  désigne un chiffre et \.  un point. Oublier l'antislash change le sens : .  seul veut dire « n'im‐

porte quel caractère ».

⚠️ PIÈGE

Corriger la zone de débordement

Dans le résultat d'une formule qui déborde, seule la première cellule contient la formule. Les autres sont grisées

dans la barre de formule et ne se modifient pas une par une.

01 Définition

EN UNE PHRASE

Il existe deux familles d'Excel : les versions perpétuelles (2016, 2019, 2021, 2024), achetées une fois

et figées, et Microsoft 365, par abonnement, qui reçoit les nouvelles fonctions en continu.

Une version perpétuelle ne reçoit jamais de nouvelle fonction : Excel 2021 ne connaîtra jamais les fonctions appa‐

rues en 2022.

02 Connaître sa version

Ouvrez Fichier > Compte. La rubrique Informations sur le produit indique « Microsoft 365 », ou le nom d'une ver‐

sion comme « Office Professionnel Plus 2021 ». Le bouton À propos d'Excel donne le numéro de build exact.

03 Les fonctions selon la version

Fonctions 2016 2019 2021 2024 Microsoft 365

SI.CONDITIONS, SI.MULTIPLE, MAX.SI.ENS, MIN.SI.ENS Non Oui Oui Oui Oui

CONCAT, JOINDRE.TEXTE Non Oui Oui Oui Oui

RECHERCHEX, EQUIVX Non Non Oui Oui Oui

FILTRE, TRIER, TRIERPAR, UNIQUE, SEQUENCE Non Non Oui Oui Oui

LET Non Non Oui Oui Oui

TEXTE.AVANT, TEXTE.APRES, FRACTIONNER.TEXTE Non Non Non Oui Oui

ASSEMB.V, ASSEMB.H, CHOISIRCOLS, LAMBDA, IMAGE Non Non Non Oui Oui

REGEXTEST, REGEXEXTRACT, REGEXREPLACE Non Non Non Non Oui

FICHE DE COURS GRATUITE

FORMULES EXCEL | VERSIONS

Les versions d'Excel

Une formule qui fonctionne parfaitement chez vous affiche

#NOM? chez un collègue : il a une version d'Excel plus

ancienne. Selon la version, des dizaines de fonctions

manquent. Cette fiche fait le point, version par version, et

indique comment vérifier la vôtre.

Version V1.0

Dernière mise à

jour
30/10/2026

Niveau 1 / 3, Découverte

Durée de lecture 8 min

Auteur Clément Delobel

Version en ligne
mesetudesdecas.fr/form
ules-excel/versions-ex
cel.html

Fonctions 2016 2019 2021 2024 Microsoft 365

GROUPER.PAR, PIVOTER.PAR Non Non Non Non Oui

Les fonctions classiques, comme SOMME, SI, RECHERCHEV, SOMME.SI.ENS ou INDEX et EQUIV, existent dans

toutes ces versions.

04 Excel pour le web et Excel sur Mac

Excel pour le web est gratuit avec un compte Microsoft. Il suit Microsoft 365 et dispose des fonctions

récentes, mais n'exécute pas les macros VBA et propose moins d'options avancées.

Excel sur Mac dispose des mêmes fonctions que la version Windows de même génération. Quelques outils

manquent ou diffèrent, comme certains compléments et une partie de l'éditeur VBA.

Excel sur mobile permet de consulter et de modifier, avec une interface simplifiée.

05 Les versions qui ne sont plus suivies

Le support de Microsoft s'est arrêté le 14 octobre 2025 pour Office 2016 et Office 2019, et le 13 octobre 2026

pour Office 2021. Ces versions continuent de fonctionner, mais ne reçoivent plus de correctif de sécurité. Office

2024 est la dernière version perpétuelle en date.

06 Face à une erreur #NOM?

Quand un fichier affiche #NOM?  dans des formules qui fonctionnaient ailleurs :

Fonction récente Équivalent dans les anciennes versions

RECHERCHEX RECHERCHEV ou INDEX + EQUIV

SI.CONDITIONS SI imbriqués

JOINDRE.TEXTE, CONCAT L'opérateur &

FILTRE, UNIQUE Le filtre avancé, Supprimer les doublons ou un tableau croisé

TEXTE.AVANT, TEXTE.APRES GAUCHE, STXT et TROUVE

💡 ASTUCE

Le vérificateur de compatibilité

Fichier > Informations > Vérifier l'absence de problèmes > Vérifier la compatibilité liste les éléments du classeur

qui ne fonctionneront pas dans une version plus ancienne. À faire avant d'envoyer un fichier à l'extérieur.

Cliquez sur la cellule et repérez la fonction dans la barre de formule.1

Vérifiez dans le tableau ci-dessus si elle existe dans votre version.2

Remplacez-la par une équivalente plus ancienne.3

07 Pièges à éviter

⚠️ PIÈGE

Le @ qui apparaît dans les formules

Un fichier créé avec Microsoft 365 et ouvert dans une ancienne version peut faire apparaître un @  devant cer‐

taines références ou noms de fonctions. Il indique un calcul ramené à une seule valeur. Inversement, des formules

anciennes ouvertes dans Microsoft 365 peuvent montrer ce @  : on peut le supprimer si la formule doit renvoyer

plusieurs valeurs.

⚠️ PIÈGE

« Office 365 » et « Microsoft 365 »

C'est le même produit : Office 365 a été renommé Microsoft 365 en 2020. Un utilisateur qui dit « j'ai Office 365 » a

donc les fonctions récentes, à condition que ses mises à jour soient faites.

⚠️ PIÈGE

Un Microsoft 365 en retard

En entreprise, le service informatique peut retarder les mises à jour de plusieurs mois. Une fonction annoncée par

Microsoft peut donc ne pas encore être disponible sur votre poste.

01 Définition

EN UNE PHRASE

L'erreur #N/A indique qu'une fonction de recherche (RECHERCHEV, RECHERCHEX, EQUIV, RECHER‐

CHEH) n'a pas trouvé la valeur cherchée dans la plage indiquée.

02 Les cinq causes et leur solution

Cause Comment la reconnaître Solution

La valeur n'existe

pas

Elle n'apparaît nulle part dans la liste Normal : masquer proprement (voir plus

bas)

Espace en trop « P-103 » contre « P-103 » suivi d'une espace SUPPRESPACE sur les deux colonnes

Nombre contre texte L'un est aligné à droite, l'autre à gauche avec un triangle

vert

Convertir en nombre, ou CNUM

Plage qui glisse Les premières lignes marchent, les suivantes non Bloquer la plage avec $ (touche F4)

Recherche

approchée

RECHERCHEV sans FAUX, sur une liste non triée Ajouter FAUX en dernier argument

03 Diagnostiquer en trente secondes

Comparez directement la valeur cherchée et celle de la liste, que vous voyez pourtant identiques :

Test de comparaison

A B C D E

1 Cherchée Dans la liste Égales ? Type cherchée Type liste

2 1045 1045 =A2=B2 =ESTTEXTE(A2) =ESTTEXTE(B2)

FICHE DE COURS GRATUITE

FORMULES EXCEL | ERREURS

L'erreur #N/A
#N/A signifie « non disponible » : Excel n'a pas trouvé ce

qu'on lui demandait de chercher. C'est l'erreur des fonctions

de recherche. Parfois c'est normal, la valeur n'existe pas.

Souvent, la valeur existe mais Excel ne la reconnaît pas, et il

faut comprendre pourquoi.

Version V1.0

Dernière mise à

jour
30/10/2026

Niveau 1 / 3, Découverte

Durée de lecture 7 min

Auteur Clément Delobel

Version en ligne
mesetudesdecas.fr/form
ules-excel/erreur-na.h
tml

Si =A2=B2  renvoie FAUX, les deux valeurs diffèrent vraiment. Les deux ESTTEXTE montrent si l'une est un texte

et l'autre un nombre. =NBCAR(A2)  révèle une espace en trop : 5 caractères au lieu de 4.

RECHERCHER EN CONVERTISSANT LA VALEUR CHERCHÉE EN TEXTE

=RECHERCHEV(A2&"";Produits!$A$2:$C$500;3;FAUX)

OU EN NOMBRE

=RECHERCHEV(A2*1;Produits!$A$2:$C$500;3;FAUX)

04 Masquer un #N/A normal

Quand l'absence est normale, par exemple un produit qui n'est pas encore au catalogue, affichez un message

clair :

AVEC RECHERCHEV

=SI.NON.DISP(RECHERCHEV(A2;Produits!$A$2:$C$500;3;FAUX);"Hors catalogue")

AVEC RECHERCHEX : C'EST INTÉGRÉ

=RECHERCHEX(A2;Produits!$A$2:$A$500;Produits!$C$2:$C$500;"Hors catalogue")

Préférez SI.NON.DISP à SIERREUR : elle ne masque que #N/A et laisse visibles les autres erreurs. Voir la fiche

SIERREUR.

05 Pièges à éviter

⚠️ PIÈGE

Masquer avant de comprendre

Ajouter SIERREUR dès le premier #N/A cache des erreurs qui n'en sont pas : un produit qui existe bien mais avec

une espace en trop sera affiché « Hors catalogue ». Cherchez la cause d'abord.

⚠️ PIÈGE

Les espaces insécables

Des données copiées depuis un site web contiennent parfois des espaces insécables, que SUPPRESPACE ne retire

pas. Utilisez =SUBSTITUE(A2;CAR(160);"") .

⚠️ PIÈGE

Le #N/A volontaire

=NA()  produit volontairement un #N/A. On l'utilise dans les graphiques : un point à #N/A n'est pas tracé, alors

qu'un zéro ferait plonger la courbe.

01 Définition

EN UNE PHRASE

L'erreur #VALEUR! apparaît quand une formule reçoit un argument d'un type qu'elle ne sait pas trai‐

ter : du texte dans une opération, une date qui n'en est pas une, des plages de tailles différentes.

02 Les causes les plus fréquentes

Cause Exemple

Du texte dans une opération =A2*2 quand A2 contient « 10 € » saisi comme texte

Un nombre stocké en texte avec une espace « 1 200 » importé d'un logiciel

Une date en texte « 30.10.2026 » ou une date d'un export anglais

Une cellule qui contient une espace Elle paraît vide, mais =A2+B2 échoue

Des plages de tailles différentes SOMME.SI.ENS(D2:D10;B2:B11;"Nord")

Une position introuvable TROUVE(" ";A2) sur un texte sans espace

FICHE DE COURS GRATUITE

FORMULES EXCEL | ERREURS

L'erreur #VALEUR!
#VALEUR! signifie qu'une formule a reçu une donnée du

mauvais type, le plus souvent du texte là où elle attendait un

nombre ou une date. La cellule fautive a souvent l'air

parfaitement normale : tout l'enjeu est de la repérer.

Version V1.0

Dernière mise à

jour
30/10/2026

Niveau 1 / 3, Découverte

Durée de lecture 7 min

Auteur Clément Delobel

Version en ligne
mesetudesdecas.fr/form
ules-excel/erreur-vale
ur.html

03 Opérateur ou fonction : pas le même comportement

Une cellule de texte dans la colonne

A B

1 Ligne Montant

2 1 120

3 2 non communiqué

4 3 80

5 Avec + =B2+B3+B4

6 Avec SOMME =SOMME(B2:B4)

L'addition avec +  renvoie #VALEUR! , car elle tente d'additionner du texte. SOMME ignore le texte et renvoie

200. Pour un total, la fonction est donc plus robuste que l'opérateur.

04 Convertir et nettoyer

Formule Ce qu'elle fait

=CNUM(A2) Convertit un texte qui ressemble à un nombre en nombre

=DATEVAL(A2) Convertit un texte qui ressemble à une date en date

=SUPPRESPACE(A2) Retire les espaces au début, à la fin et en double

=SUBSTITUE(A2;" €";"") Retire un symbole avant conversion

=SUBSTITUE(A2;".";"/") Transforme 30.10.2026 en 30/10/2026

💡 ASTUCE

Sans formule : Convertir

Pour une colonne entière de dates ou de nombres en texte : sélectionnez-la, puis Données > Convertir, cliquez di‐

rectement sur Terminer. Excel réinterprète chaque cellule. Pour des dates au format inhabituel, choisissez le format

à l'étape 3.

05 Trouver la cellule fautive

Cliquez sur la cellule en erreur, puis sur le losange d'avertissement qui apparaît à côté.1

Choisissez Afficher les étapes du calcul : l'évaluation de formule montre à quel moment l'erreur apparaît.2

Ou sélectionnez chaque morceau de la formule dans la barre de formule et appuyez sur F9  pour voir sa

valeur (puis Échap ). Voir la fiche assistant fonction.

3

06 Pièges à éviter

⚠️ PIÈGE

Le format ne convertit rien

Appliquer le format Nombre ou Date à une cellule qui contient du texte ne change rien : la donnée reste du texte. Il

faut la convertir, ou la ressaisir.

⚠️ PIÈGE

Le séparateur décimal

Un fichier exporté avec des points décimaux, comme 12.50, est lu comme du texte dans un Excel en français. Rem‐

placez les points par des virgules avec Rechercher et remplacer ( Ctrl + H ) avant de calculer.

⚠️ PIÈGE

Masquer avec SIERREUR

=SIERREUR(A2*2;0)  fait disparaître l'erreur, mais le résultat devient faux : 0 au lieu du vrai montant. Corrigez la

donnée plutôt que la formule.

